



# **UNIVERSIDAD NACIONAL DE AVELLANEDA**

**DPTO. DE TECNOLOGÍA Y ADMINISTRACIÓN  
INGENIERÍA EN INFORMÁTICA**

Sistemas de Control Automático

## **Trabajo de Laboratorio 2**

**Control Lineal de un Ascensor**

ESTUDIANTES

**Massimino, Matías**

**Colli, Alejandro**

**Eroles, Pedro**

DOCENTE

**Guillermo Federico Caporaletti**

2do Cuatrimestre 2023

# Índice

<b>Índice</b> .....	2
<b>Objetivo</b> .....	3
<b>Introducción</b> .....	4
<b>Modelo del Ascensor</b> .....	5
<b>Equipamiento y componentes utilizados</b> .....	12
<b>Desarrollo</b> .....	14
<b>Mediciones</b> .....	16
<b>Conclusión</b> .....	19
<b>Mejoras a desarrollar</b> .....	20
<b>Anexo</b> .....	21
<b>Bibliografía</b> .....	35

## Objetivo

Nuestro objetivo principal es controlar el sistema físico mediante la acción de un sistema de control de tipo Proporcional Derivativo. Se continuó utilizando el equipo utilizado en la práctica anterior.

La experiencia debe ser registrada para poder medir la respuesta del sistema y elegir los parámetros más adecuados para configurar nuestro controlador teniendo en cuenta: tiempo de respuesta, precisión y sobrevalor. Debemos evaluar su comportamiento real a partir de la experiencia.

Para eso, se va a proceder a modificar el código existente que se utilizó en la anterior presentación, agregando el control PWM, obteniendo el tiempo de derivación (TD) y la constante de proporcionalidad o ganancia (KP).

## Introducción

En este trabajo vamos estar utilizando un sistema de lazo cerrado o realimentado, en el cual exploraremos los principios teóricos detrás del control proporcional derivativo y su implementación práctica para controlar nuestro sistema físico. Se llevarán a cabo experimentos con distintas pruebas y análisis para evaluar la respuesta transitoria, estabilidad, sobrevalor y eficiencia del sistema.

Posteriormente, en XCos se harán las simulaciones correspondientes, las cuales se irá variando tanto variando la ganancia como el derivativo, encontrando así la ganancia que más se adecue a nuestras necesidades.

Por último, con estos valores obtenidos, se procederá a realizar la experiencia de laboratorio.

## Modelo del Ascensor

Para la resolución del siguiente Trabajo Práctico, lo primero que se debe realizar es un diagrama en bloques que represente al sistema con el cual se trabaja; como así también la identificación de la planta, la variable controlada y la variable manipulada.




Figura 1: Diagrama en bloques del sistema.

El sistema que estamos controlando se puede representar en un diagrama en bloques, como se observa en la figura anterior. En nuestro sistema, la planta va a ser el ascensor, el cual va a contener un motor y un optoacoplador; la variable controlada va a ser la altura del ascensor; y finalmente la variable manipulada va a ser el voltaje que reciba el motor para subir el ascensor.

En nuestro sistema, la acción de control es del PD, en el cual se utiliza todo el rango de la señal manipulada. En nuestro caso de 0 a 7V. para controlar la velocidad de la planta.

Esto se hace multiplicando la señal de error por un coeficiente de proporcionalidad ( $K_p$ ) y obteniendo con esto la corrección del sistema.

## Simulaciones

Antes de comenzar con el desarrollo, realizamos algunas simulaciones utilizando Scilab. Como ya teníamos el modelo desarrollado en la práctica anterior, utilizamos la transferencia obtenida para realizar las simulaciones.





Figura 2: Simulación control PD frente al escalón.

En esta simulación, lo que tenemos es el control proporcional derivativo del ascensor frente a un escalón. Tuvimos en cuenta la zona muerta, el rozamiento y la saturación, por eso es que utilizamos una ganancia de 1,2.

Los resultados de la simulación fueron los siguientes.



*Figura 3: Respuesta al escalón*

Obtuvimos un tiempo de establecimiento de  $1,55\text{ s}$  y un tiempo de crecimiento de  $1,04\text{ s}$ . El retardo es de  $0,4$

```

1  //Ejemplo de control integral
2
3  //Función de transferencia y espacio de simulación
4  s=s
5  t=0:0.01:10
6  G=-5.93/s/(1+0.221*s)
7  G=syslin('c',G)
8
9  KP = 0.6
10
11
12
13 close()
14 close()
15 close()
16 figure(1)
17 evans(G*(1+s*0.4),10)
18 //evans(G*(1+s*0.2),10)
19 plot([0,-5],[0,5],"--")
20
21
22 //Pruebas variando el TD
23 figure(2)
24 for TD=[0,-0.1,0.2,0.4]
25     PD = KP.*(1+s*TD)
26     LC=(PD*G)/(1+PD*G)
27     Y=csim('step',t,LC);
28     plot(t,Y);
29 end
30
31 //Respuesta frente a una entrada rampa
32
33 rampa=t*5;
34 for i=-201:1:1001
35     rampa(i)=10;
36 end
37
38 figure(3)
39 plot(t,rampa,"r");
40 for TD=[0,0.1,0.2,0.4]
41     PD = KP.*(1+s*TD)
42     LC=(PD*G)/(1+PD*G)
43     Y=csim(rampa,t,LC);
44     plot(t,Y);
45 end

```

Figura 4: Script para simulaciones

En esta segunda simulación lo que tenemos es un Script con el control proporcional derivativo del Ascensor.

Definimos la función de transferencia, y un  $K_p$  de 0,6. La figura 1 nos muestra el Evans.



La figura 2 nos muestra algunas pruebas variando el  $TD$  frente a una entrada escalón.

La figura 3 nos muestra la respuesta frente a una entrada rampa.

Como resumen, este código se utiliza para realizar simulaciones y análisis de un controlador proporcional derivativo en un sistema representado por la función de transferencia. Analizamos la respuesta para varios valores de  $TD$  y evaluamos el comportamiento frente a entradas de escalón y rampa.






Figura 5: Diagrama de Evans



*Figura 6: Respuesta al escalón*



*Figura 7: Respuesta frente a una rampa*

## Equipamiento y componentes utilizados


Para esta práctica de laboratorio utilizamos los mismos componentes que en la práctica anterior, pero esta vez sumando el uso del PWM.

- Puente H
- Fuente de Tensión
- Pulsador
- Arduino Uno
- Optoacoplador

### PWM

PWM son las siglas de Pulse Width Modulation (Modulación por ancho de pulso). Para transmitir una señal, ya sea analógica o digital, se debe modular para que sea transmitida sin perder potencia o sufrir distorsión por interferencias.

PWM es una técnica que se usa para transmitir señales analógicas cuya señal portadora será digital. En esta técnica se modifica el ciclo de trabajo de una señal periódica (una senoidal o una cuadrada, por ejemplo), ya sea para transmitir información a través de un canal de comunicaciones o para controlar la cantidad de energía que se envía a una carga.



*Figura 8: PWM*

Esta modulación es muy usada para controlar la cantidad de energía que se envía a una carga, es una técnica utilizada para regular la velocidad de giro de los motores, regulación de intensidad luminosa, controles de elementos termoelectrónicos o controlar fuentes conmutadas entre otros usos.

La mayoría de los automatismos, incluido Arduino, no son capaces de proporcionar una señal analógica. Sólo pueden proporcionar una salida digital de  $-V_{cc}$  o  $V_{cc}$ . (por ejemplo,  $0V$  y  $5V$ ). Entonces, para conseguir una señal analógica, la mayoría de los automatismos usan PWM. Se usa esta técnica porque como se ve en los ejemplos anteriores, no siempre quieres un valor digital de la señal (ON/OFF), si no que necesitaremos proporcionar un valor analógico de tensión que usarán para las aplicaciones deseadas.

En nuestro caso, utilizamos el PWM para controlar la potencia aplicada al motor, cuando el ascensor está subiendo ajustamos la potencia aplicada al motor para moverlo en función de la altura objetivo, y cuando el ascensor está bajando se utiliza de manera similar.

## Desarrollo

Como ya teníamos la implementación realizada de la práctica anterior, lo único que nos quedaba hacer era volver a realizar las conexiones, y agregar la conexión para el funcionamiento del PWM.


Esta vez, pudimos utilizar el pulsador para detener el sistema, cosa que en la anterior practica no habíamos podido realizar por falta de tiempo.



*Figura 9: Conexionado del Ascensor*

Luego de volver a realizar todo el conexionado, nos pusimos a modificar el código para que pueda funcionar nuestro sistema con un control proporcional derivativo.

Para el desarrollo de nuestro proyecto, contamos con un control proporcional derivativo, que a partir de las funciones “controlar” y “actuar” definidas en el código, decidimos si el motor debe subir, bajar, ir deteniéndose o detenerse en función de la posición en la que se encuentre y la posición establecida como objetivo.



*Figura 10: Pruebas de las conexiones*

## Mediciones

Ya habiendo hecho el código y las conexiones de nuestro sistema, podemos pasar a la experiencia en sí. Lo que hicimos fue realizar varias mediciones, probando con distintos valores de  $Kp$  y  $Td$  para poder encontrar los valores que dieran la mayor precisión de nuestro sistema.

Nuestras mediciones las realizamos cada 50  $ms$ . Específicamente, para la primera medición establecimos una altura objetivo de 10  $cm$ , con un  $Kp$  de 1,0 y un  $Td$  de 0,2.




Figura 11: Resultados de la primera medición.



Como podemos observar, vemos que el ascensor llega hasta los  $9,76\text{ cm}$  y se detiene, lo cual indica una muy buena precisión, ya que el ascensor no sigue de largo y logra detenerse.

Para la segunda medición, lo que hicimos fue definir en nuestro código 3 pisos para el ascensor, con altura de  $22\text{ cm}$  para el piso 3,  $11\text{ cm}$  para el piso 2, y  $0\text{ cm}$  para el piso 1. Para esto, utilizamos un  $K_p$  de  $2,6$  y un  $T_d$  de  $0,2$ .




Figura 12: Resultados de la segunda medición

Como podemos observar, el ascensor se comporta de muy buena manera, ya que llega a sus 3 pisos objetivo sin pasarse de largo y deteniéndose cuando debe hacerlo.

Finalmente, para la tercer y última medición lo que hicimos fue utilizar un control proporcional integral. Utilizamos un  $K_p$  de  $3,2$ , un  $T_d$  de  $0,25$  y un  $T_i$  de  $2$ .

Los resultados obtenidos fueron los siguientes




Figura 13: Resultados de la tercera medición

Una vez hechas las 3 mediciones, podemos afirmar que el tiempo de establecimiento es de  $1300\text{ ms} = 1,3\text{ s}$  lo cual es muy parecido a lo que habíamos estimado en las simulaciones hechas.

## Conclusión

Como cierre, podemos decir que considerando nuestro objetivo principal al momento de hacer la practica era controlar el ascensor, logramos concluir el trabajo de laboratorio con éxito, haciendo que nuestro control proporcional derivativo funcione correctamente de acuerdo a los valores utilizados y la altura medida y el objetivo determinado. Pudimos comprobar la muy buena precisión que tiene nuestro sistema con la correcta elección de los tiempos de derivación y las constantes de proporcionalidad.

Esta vez, pudimos utilizar el pulsador para detener el sistema, y pudimos comprobar su correcto funcionamiento a la hora de presionarlo.

En comparación con la practica anterior, podemos decir que esta práctica se hizo más sencilla, ya que teníamos armado el ascensor, ya sabíamos cómo funcionaba el optoacoplador y los demás componentes, y en base al código esta vez al modularizar todo, fue mucho más claro y fácil de entender.

Afortunadamente, superamos estos desafíos y conseguimos completar con éxito nuestro trabajo de laboratorio, lo que dejó a todos nosotros muy satisfechos.

## Mejoras a desarrollar

Una mejora por realizar sería principalmente la posibilidad de agregar un tiempo de integración, un control integral además del control proporcional derivativo que se utiliza actualmente, este control integral podría darle mayor precisión a los desplazamientos u objetivos que tiene el ascensor a cada movimiento, reduciendo oscilaciones y dando mayor estabilidad, especialmente al momento de perturbaciones o cambios en las condiciones de la práctica. Durante la práctica se estuvieron haciendo experimentos con este control integral, pero el resultado no fue el esperado, debido a la disminución del voltaje que tiene en ese momento, no puede desplazarse correctamente para mejorar la precisión y no consigue terminar la acción, se tendría que verificar si es posible controlar mejor el voltaje para poder utilizarlo.

Otra mejora que se podría implementar, pero esta vez a un nivel físico del sistema, podría ser la incorporación de finales de carrera, tanto para el piso inferior como para el superior, así mismo se podría seguir agregar distintos pisos al ascensor. Por otro lado, que tanto el peso del ascensor como el del contrapeso sean el mismo, para que la medición obtenga una mayor precisión.

## Anexo

A continuación, se coloca el código fuente elaborado para el control proporcional derivativo que se usó para el ascensor, con sus respectivos módulos.

```

1  #include "Posicion.h"
2  #include "ControlPID.h"
3
4  // Definicion de pines
5  #define PIN_PH1      8    // Control para el pin 1 del motor A // Baja al alimentar el IN1
6  #define PIN_PH2      9    // Control para el pin 2 del motor A // Sube el motor al alimentar IN2
7  #define PIN_PH_PWM   10   // Elegir un pin que admita PWM //
8  #define PIN_EMERGENCIA 11  // Defino un pin para leer el boton de emergencia
9  #define PIN_LED       13   // Led de emergencia
10 #define PIN_OPTOACOPLADOR 12 // Elegir //Defino un pin para el optoacoplador
11
12 // Definicion de tiempos
13 #define DELTA_TIEMPO  50   // Intervalo entre muestras para control
14 #define INTERVALO_CONTROL 50 // en milisegundos
15 #define MUESTRAS_PARA_DETENER 6
16
17 // Constantes de parámetros de control PID
18 const float kp = 1.0; // Ganancia proporcional
19 const float ti = 0; // Ganancia integral (puedes ajustar según sea necesario)
20 const float td = 0.2; // Ganancia derivativa
21
22 // Variables globales
23 controlPID controlAscensor (kp, ti, td); // Instancia de la clase ControlPID
24 float altura_objetivo = 10;
25 float altura_medida = 0;
26 float accion_deseada_pwm;
27 const float altura_piso[3] = {0,11,22};
28 estado_motor accion_deseada = DETENIDO;
29
30 // Se establece una variable ALTURA en 0 (la cual será el destino del ascensor)
31 // Definimos una variable MOTOR que nos indique 0 si esta quieto, 1 si sube y 2 si baja
32 // Se establece una variable ALT_OBJ en 0
33 // El ascensor estará en la planta baja.
34 // float ALTURA = 0.0; // unificar unidades de altura y tipo de variable
35 //estado_motor accion_motor = DETENIDO; // Estado inicial del motor
36
37 unsigned long ultimo_tiempo = 0;
38 unsigned long tiempo_actual = 0;
39 unsigned long tiempo_inicial = 0;
40
41 // Declaración de funciones
42 void mostrar();
43 void detenerMaquina();
44 void controlar();
45 void actuar();

```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
49 void setup() {
50     inicializarPosicion(PIN_OPTOACOPLADOR);
51     pinMode(PIN_PH1, OUTPUT); // Se inician los pines
52     pinMode(PIN_PH2, OUTPUT);
53     pinMode(PIN_EMERGENCIA, INPUT);
54     pinMode(PIN_PH_PWM, OUTPUT);
55     pinMode(PIN_LED, OUTPUT);
56
57     controlAscensor.LimitarSalida(true, -TENSION_MAXIMA, TENSION_MAXIMA);
58
59     // Inicialicen la conexión serial
60     Serial.begin(115200); // poner velocidad mas rapida
61
62     // Encabezado de datos:
63     Serial.println("Encabezado de datos");
64     Serial.println("Tiempo\tAlt \tObj \tProp \tDer \tAccion");
65     Serial.println("[ms] \t[cm] \t[cm] \t[V] \t[V] \t[V]");
66
67     // Antes de comenzar
68     delay(3000); // Luego de 3 segundos arrancamos el programa.
69     ultimo_tiempo = millis();
70     tiempo_actual = ultimo_tiempo;
71     tiempo_inicial = ultimo_tiempo;
72     mostrar();
73 }
74
75 void loop() {
76     tiempo_actual = millis();
77
78     // Verificamos el botón de emergencia (PIN_EMERGENCIA)
79     if (digitalRead(PIN_EMERGENCIA) == HIGH) {
80         detenerMaquina(); // Realizamos la parada de emergencia
81     }
82
83     // Obtenemos la altura medida utilizando la función de Posicion
84     altura_medida = leerPosicion(accion_deseada);
85
86     if (millis() - ultimo_tiempo >= DELTA_TIEMPO) {
87         ultimo_tiempo += DELTA_TIEMPO;
88         // Analizamos altura objetivo
89         if (ultimo_tiempo - tiempo_inicial < 4000) {
90             altura_objetivo = altura_piso[2];
91         } else if (ultimo_tiempo - tiempo_inicial < 8000) {
92             altura_objetivo = altura_piso[1];
93         } else if (ultimo_tiempo - tiempo_inicial < 12000) {
94             altura_objetivo = altura_piso[0];
95         }
96         // Llamamos a la función de control
97         accion_deseada_pwm = controlAscensor.Controlar(altura_objetivo - altura_medida);
98
99         // Llamada a la función de actuación
100         actuar();
101
102         // Mostramos información
103         mostrar();
104     }
105 }
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
113 // Parada de emergencia
114 void detenerMaquina() {
115     // Lógica para detener el ascensor de emergencia
116     digitalWrite(PIN_PH1, LOW);
117     digitalWrite(PIN_PH2, LOW);
118     analogWrite(PIN_PH_PWM, 0);
119     digitalWrite(PIN_LED, HIGH);
120     Serial.println("Parada de emergencia");
121
122     do {} while (1);
123 }

127 void actuar()
128 {
129     int pwmValue = map(abs(accion_deseada_pwm), TENSION_MIN_ACT, TENSION_MAXIMA, 0, 255);
130     if(pwmValue == TENSION_MIN_ACT){
131         pwmValue = 0;
132     }
133     static int cuenta = 0;
134     static int cuenta2 = 0;
135     switch (accion_deseada){
136     case DETENIDO:
137         cuenta2 = 0;
138         // Ascensor Detenido
139         if (altura_objetivo - altura_medida > TOLERANCIA ){
140             accion_deseada = SUBIENDO;
141         }
142         if (altura_medida - altura_objetivo > TOLERANCIA){
143             accion_deseada = BAJANDO;
144         }
145         break;
146
147     case SUBIENDO:
148         // Ascensor Subiendo
149         if(accion_deseada_pwm <= 0){
150             accion_deseada = DETENIENDO_SUB;
151         }
152         if(abs(altura_objetivo - altura_medida) <= TOLERANCIA){
153             cuenta2++;
154             if(cuenta2 > MUESTRAS_PARA_DETENER){
155                 accion_deseada = DETENIDO;
156             }
157         }
158         break;
159
160     case BAJANDO:
161         // Ascensor Bajando
162         if(accion_deseada_pwm >= 0){
163             accion_deseada = DETENIENDO_BAJ;
164         }
165         if(abs(altura_objetivo - altura_medida) <= TOLERANCIA){
166             cuenta2++;
167             if(cuenta2 > MUESTRAS_PARA_DETENER){
168                 accion_deseada = DETENIDO;
169             }
170         }
171         break;
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
173     case DETENIENDO_SUB:
174         cuenta2 = 0;
175         // Ascensor Deteniendose
176         cuenta++;
177         if ((abs(LeerVelocidad())) < VELOCIDAD_MINIMA){
178             accion_deseada = DETENIDO;
179             cuenta=0;
180         }
181         if (cuenta > MUESTRAS_PARA_DETENER) {
182             accion_deseada = DETENIDO;
183             cuenta=0;
184         }
185         if(accion_deseada_pwm > 0){
186             accion_deseada = SUBIENDO;
187         }
188         break;
189
190     case DETENIENDO_BAJ:
191         cuenta2 = 0;
192         // Ascensor Deteniendose
193         cuenta++;
194         if ((abs(LeerVelocidad())) < VELOCIDAD_MINIMA){
195             accion_deseada = DETENIDO;
196             cuenta=0;
197         }
198         if (cuenta > MUESTRAS_PARA_DETENER) {
199             accion_deseada = DETENIDO;
200             cuenta=0;
201         }
202         if(accion_deseada_pwm < 0){
203             accion_deseada = BAJANDO;
204         }
205         break;
206     }
207 }
```



## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
208 switch (accion_deseada){
209
210     case DETENIDO:
211         // Ascensor Detenido
212         analogWrite(PIN_PH_PWM, 0);
213         accion_deseada_pwm = 0;
214         break;
215
216     case SUBIENDO:
217         // Ascensor Subiendo
218         digitalWrite(PIN_PH1, LOW);
219         digitalWrite(PIN_PH2, HIGH);
220         analogWrite(PIN_PH_PWM, pwmValue);
221         break;
222
223     case BAJANDO:
224         // Ascensor Bajando
225         digitalWrite(PIN_PH2, LOW);
226         digitalWrite(PIN_PH1, HIGH);
227         analogWrite(PIN_PH_PWM, pwmValue);
228         break;
229
230     case DETENIENDO_SUB:
231         // Ascensor Deteniendose
232         digitalWrite(PIN_PH2, LOW);
233         digitalWrite(PIN_PH1, LOW);
234         analogWrite(PIN_PH_PWM, 0);
235         accion_deseada_pwm = 0;
236         break;
237
238     case DETENIENDO_BAJ:
239         // Ascensor Deteniendose
240         digitalWrite(PIN_PH2, LOW);
241         digitalWrite(PIN_PH1, LOW);
242         analogWrite(PIN_PH_PWM, 0);
243         accion_deseada_pwm = 0;
244         break;
245 }
246 }
```

```

248 void mostrar() {
249     // Imprimimos los valores de las variables y del controlador PID
250     //Serial.print("Tiempo: ");
251     Serial.print(ultimo_tiempo - tiempo_inicial);
252     Serial.print("\t");
253
254     //Serial.print("ALTURA actual: ");
255     Serial.print(altura_medida); // Cambiado de ALTURA a altura_medida
256     Serial.print("\t");
257
258     //Serial.print("ALTURA objetivo: ");
259     Serial.print(altura_objetivo);
260     Serial.print("\t");
261
262     //Serial.print("Acción Proporcional: ");
263     Serial.print(controlAscensor.ObtenerProporcional());
264     Serial.print("\t");
265
266     //Serial.print("Acción Derivativa: ");
267     Serial.print(controlAscensor.ObtenerDerivativo());
268     Serial.print("\t");
269
270     //Serial.print("Salida Controlador PID: ");
271     Serial.print(accion_deseada_pwm);
272     Serial.print("\t");
273
274     switch (accion_deseada) {
275     case SUBIENDO:
276         Serial.println("SUB");
277         break;
278     case DETENIDO:
279         Serial.println("DET");
280         break;
281     case DETENIENDO_SUB:
282         Serial.println("DNS");
283         break;
284     case BAJANDO:
285         Serial.println("BAJ");
286         break;
287     case DETENIENDO_BAJ:
288         Serial.println("DNB");
289         break;
290     }
291 }

```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
1 #ifndef ASCENSOR_H
2 #define ASCENSOR_H
3
4 #define INTERVALO_MINIMO 5 // Intervalo mínimo entre cambios de estados, en milisegundos.
5 #define DISTANCIA 0.287 // Distancia de un cambio de estado del optoacoplador, en cm.
6 // Cambios de estados y desplazamiento:
7 // La polea tiene 46 cambios de estados en una vuelta.
8 // Diametro de polea = 42 mm => Recorrido en una vuelta pi*diametro = 131 mm
9 // Cada cambio de estado = 131 mm / 46 = 2,87 mm = 0,287 cm
10 #define TOLERANCIA 1.0 // Define la tolerancia en centímetros
11 #define VELOCIDAD_MINIMA 1.0 // Define la velocidad mínima en cm/s
12 #define VEL_MAXIMA 35.6 // cm/s
13 #define VELOCIDAD_CONVERSION 5.93 // cm/V/s
14 #define TENSION_MAXIMA 7
15 #define TENSION_MIN_ACT 3
16
17 typedef enum { SUBIENDO, BAJANDO, DETENIENDO_SUB, DETENIENDO_BAJ, DETENIDO } estado_motor;
18
19 estado_motor leerUltimoMovimiento();
20 void inicializarPosicion (int pin_opto);
21 float leerPosicion (estado_motor accion_deseada);
22 float leerVelocidad ();
23
24 #endif
25
```

```
1 #include "Posicion.h"
2 #include "Arduino.h"
3
4 estado_motor ultima_accion = DETENIDO;
5 unsigned long tiempo_ultimo_cambio_0 = 0;
6 unsigned long tiempo_ultimo_cambio_1 = 0;
7 unsigned long tiempo_ultimo_cambio_2 = 0;
8 bool estadoOptoacoplador;
9 float altura = 0;
10 int pin_optoacoplador;
11
12 //*****
13 void inicializarPosicion(int pin_opto)
14 {
15     pin_optoacoplador = pin_opto;
16     // Configuramos el pin del optoacoplador como entrada
17     pinMode(pin_optoacoplador, INPUT);
18     estadoOptoacoplador = digitalRead(pin_optoacoplador);
19     altura = 0;
20 }
21
22 //*****
23 estado_motor leerEstadoMotor()
24 {
25     return ultima_accion;
26 }
27
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
29 float leerPosicion(estado_motor accion_deseada)
30 {
31
32     // Obtenemos el tiempo actual y leo estado de optoacoplador
33     unsigned long tiempo_actual = millis();
34     bool lectura_actual = digitalRead(pin_optoacoplador);
35
36     // Verificamos si hubo cambio de estado
37     if (estadoOptoacoplador != lectura_actual) {
38
39         // ANTIRREBOTE: verificamos si ha pasado un tiempo mínimo
40         if (tiempo_actual - tiempo_ultimo_cambio_0 >= INTERVALO_MINIMO) {
41             // Hubo un cambio de estado verificado!!!
42             estadoOptoacoplador = lectura_actual;
43
44             // Guardamos los tiempos de los dos últimos cambios de estado
45             tiempo_ultimo_cambio_2 = tiempo_ultimo_cambio_1;
46             tiempo_ultimo_cambio_1 = tiempo_ultimo_cambio_0;
47             tiempo_ultimo_cambio_0 = tiempo_actual;
48
49             // Actualizar la última acción si el motor estaba subiendo o bajando
50             if (accion_deseada == SUBIENDO) {
51                 ultima_accion = SUBIENDO;
52             }
53             if (accion_deseada == BAJANDO) {
54                 ultima_accion = BAJANDO;
55             }
56
57             // Calculamos la posición actual
58             if (ultima_accion==SUBIENDO) {altura+=DISTANCIA;};
59             if (ultima_accion==BAJANDO) {altura-=DISTANCIA;};
60         }
61     }
62
63     //Devolvemos la altura
64     return altura;
65 }
66
67
68 float leerVelocidad()
69 {
70     float medicion = 0;
71     unsigned int delta_actual = 0; // es el delta actual, que mide si reduce la velocidad o no
72     unsigned int delta_anterior = 0; // es el delta anterior, medido entre los deltas de tiempo actual y anterior al anterior.
73
74     if (0 == tiempo_ultimo_cambio_1){
75         // Por el momento no hubo ninguna interrupción almacenada
76         // Se supone una velocidad de 0 RMP.
77         medicion = 0;
78     } else {
79         delta_actual = millis() - tiempo_ultimo_cambio_1;
80         delta_anterior = tiempo_ultimo_cambio_0 - tiempo_ultimo_cambio_2;
81
82         if (0 != tiempo_ultimo_cambio_2) {
83             // Verifico que ya habia sido actualizado tiempo_ultimo_cambio_2
84             // y elijo la menor velocidad (== a el mayor tiempo):
85             delta_actual = max(delta_actual, delta_anterior);
86         }
87
88         medicion = 1000 * 2 * DISTANCIA / (delta_actual);
89     }
90
91     return medicion;
92 }
93
94 estado_motor leerUltimoMovimiento()
95 {
96     return ultima_accion;
97 }
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
1  /*****
2  * Control PID
3  *****/
4  * Sistemas de Control Automatico (SCA)
5  * Universidad Nacional de Avellaneda (UNDAV)
6  *
7  * Archivo:   pid_sca.h
8  * Version:   2.1.
9  * Fecha:     Noviembre 2023
10 * Novedades: Reorganizacion de funciones en .h y .cpp
11 * Version Anterior: 2.0, diciembre 2021.
12 *****/
13
14 #define COHEFICIENTE_FILTRO 0.6
15
16 class controlPID          // Objeto para control Proporcional-Integral-Derivativo (PID)
17 {
18 private:
19
20     float Salida;          // La señal de control que va al acuator
21     // o potencia de salida (sin asignar unidades)
22     float Proporcional;    // Componente proporcional de la salida
23     // (sin asignar unidades)
24     float Integral;        // Componente integral
25     float Derivativo;      // Componente derivativa
26     float Compensacion;    // Contiene la compensacion resultante ante saturacion
27     float CompensacionAnterior; // Como usamos aproximacion trapezoidal de la integral,
28     // necesitamos conservar el valor anterior.
29     float Kp;              // Constante proporcional (sin asignar unidades)
30     float Ti;              // Tiempo de integracion (en segundos)
31     float Td;              // Tiempo para la componente derivativa (en segundos)
32     unsigned long TiempoAnterior; // Tiempo de la medicion anterior utilizando micros
33     float ErrorAnterior;   // Señal de error anterior
34     bool LimitaSalida;     // Indica si establecimos limites superior e inferior
35     bool CompensaIntegral; // Indica si establecimos la compensacion de integral
36     float SalidaMax;       // Limite superior de la salida (y de la integral)
37     float SalidaMin;       // Limite inferior de la salida
38     const float MILLON = 1e6; // Constante para convertir micros() a segundos.
39 }
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
40 public:
41
42     // Constructor con lo minimo:
43     controlPID(float KP, // KP: Constante de proporcionalidad (puede ser negativo)
44                 float TI, // TI: Tiempo de integracion (si es 0, no integra)
45                 float TD); // TD: Tiempo de derivacion (si es 0 no deriva)
46
47     // Para cambiar configuracion inicial:
48     void ConfigurarPID(float KP, float TI, float TD); //Mismos parametros que constructor.
49
50     // Configura los limites de salida e indica si estan activados:
51     bool LimitarSalida(bool RESPUESTA, float SMIN, float SMAX);
52
53     // Activa o desactiva los limites de salida e indica si el limite de salida esta activado:
54     bool LimitarSalida(bool RESPUESTA); // No permite activar limites si antes no fueron
55     // establecidos.
56
57     // Indica si el limite de salida esta activado:
58     bool LimitarSalida();
59
60     // Activa o desactiva la compensacion de integracion e indica si esta activado:
61     bool CompensarIntegral(bool RESPUESTA);
62
63     // Me indica si la compensacion esta activada:
64     bool CompensarIntegral();
65
66     // Calcula señal de control (salida) en funcion del error:
67     float Controlar(float ERROR);
68
69     // Apaga el PID manteniendo configuracion:
70     void Apagar(); // No se modifican los valores de KP, TI y TD.
71     // Tampoco los limites pre establecidos.
72
73     float ObtenerIntegral();
74     float ObtenerProporcional();
75     float ObtenerDerivativo();
76     float ObtenerSalida();
77     float ObtenerCompensacion();
78
79 };
80
81 1  /*****
82 2  * Control PID
83 3  *****/
84 4  * Sistemas de Control Automatico (SCA)
85 5  * Universidad Nacional de Avellaneda (UNDAV)
86 6  *
87 7  * Archivo:    pid_sca.cpp
88 8  * Version:    2.1.
89 9  * Fecha:      Noviembre 2023
90 10 * Novedades:  Reorganizacion de funciones en .h y .cpp
91 11 * Version Anterior: 2.0, diciembre 2021.
92 12 *****/
93 13
94 14 #include "ControlPID.h"
95 15 #include "Arduino.h"
96 16
97 17 /*****/
98 18
99 19 controlPID::controlPID(float KP, float TI, float TD)
100 // Constructor: incluye configuracion inicial del PID y valores predeterminados.
101 { // Kp puede ser negativo (esto ultimo podria servir para controlar una planta cuya salida
102   // tienda a bajar cuando aumente la señal de control. Ej.: heladera.)
103   // Si Ti=0, el PID no lo tomar en cuenta
104   // Si Td=0, el PID no lo tomar en cuenta
105   // Inicializa integracion e impone false en limites y compensacion.
106   ConfigurarPID(KP, TI, TD);
107   LimitaSalida = false;
108   CompensaIntegral = false;
109 }
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
32 void controlPID::ConfigurarPID(float KP, float TI, float TD)
33 // Configura las constantes basicas del control PID
34 // Sirve para cambiar configuracion inicial, sin modificar limites y banderas.
35 { // Kp puede ser negativo (esto ultimo podria servir para controlar una planta cuya salida
36   //                               tienda a bajar cuando aumente la señal de control. Ej.: heladera.)
37   // Si Ti=0, el PID no lo tomar en cuenta
38   // Si Td=0, el PID no lo tomar en cuenta
39   Kp = KP;
40   Ti = TI;
41   Td = TD;
42   // Resetea valores de integracion.
43   TiempoAnterior = 0;
44   ErrorAnterior = 0;
45   CompensacionAnterior = 0;
46   Integral = 0;
47 }
48 //-----
49
50 bool controlPID::LimitarSalida()
51 // Devuelve el valor de la variable privada LimitaSalida
52 // que indica si nuestro PID esta configurado para limitar su salida.
53 {
54   return LimitaSalida;
55 }
56 //-----
57
58 bool controlPID::LimitarSalida(bool RESPUESTA)
59 // Configura si limitar la salida...
60 // No permite activar limites si antes no fueron establecidos.
61 {
62   LimitaSalida = RESPUESTA;
63   if (SalidaMax == 0 && SalidaMin == 0) {
64     // ...no voy a limitar porque no tengo limites definidos.
65     LimitaSalida = false;
66   }
67   return LimitaSalida;
68 }
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
71 bool controlPID::LimitarSalida(bool RESPUESTA, float SMIN, float SMAX)
72 // Configura si limitar la salida entre SMAX y SMIN.
73 // Se puede establecer los limites pero no activarlos aun.
74 // No activa con SMIN=SMAX.
75 // No activa si SMIN>SMAX.
76 {
77     SalidaMax = SMAX;
78     SalidaMin = SMIN;
79     LimitaSalida = RESPUESTA;
80     // La forma de desactivar este limite es:
81     // 1) Volviendo a llamar esta funcion con RESPUESTA=false
82     // 2) Llamando a LimitarSalida(false)
83     // Tambien se pueden poner limites muy grandes.
84     if (SalidaMax == SalidaMin) {
85         // ...no voy a limitar porque no tengo limites definidos.
86         LimitaSalida = false;
87     }
88     if (SalidaMin > SalidaMax) {
89         // ...no voy a limitar porque estan mal configurados.
90         LimitaSalida = false;
91     }
92     return LimitaSalida;
93 }
94 //-----
95
96 bool controlPID::CompensarIntegral()
97 // Devuelve el valor de CompensaIntegral.
98 {
99     return CompensaIntegral;
100 }
101 //-----
102
103 bool controlPID::CompensarIntegral(bool RESPUESTA)
104 // Establece si debo compensar la integracion cuando la salida esta saturada.
105 // Deben haberse preestablecido los limites de salida.
106 {
107     CompensaIntegral = RESPUESTA;
108     if (!LimitaSalida) CompensaIntegral = false;
109     return CompensaIntegral;
110 }
```



## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
113 float controlPID::Controlar(float ERROR)
114 // Calcula Salida en funcion de la señal error y los parametros del PID
115 {
116     unsigned long TiempoActual = micros(); // Tomo tiempo actual para comparar con anterior
117
118     // PROPORCIONAL -----
119     Proporcional = Kp * ERROR;
120
121     // DERIVATIVO -----
122     if (TiempoAnterior > 0 && Td != 0) {
123         // Dos condiciones para componente derivativa:
124         // 1) Que no sea el primer calculo y 2) Td seteado
125         float Cuenta = Kp * Td * (ERROR - ErrorAnterior) * MILLON / (TiempoActual - TiempoAnterior);
126         Derivativo = COHEFICIENTE_FILTRO * Cuenta + (1-COHEFICIENTE_FILTRO) * Derivativo;
127     }
128     else {
129         Derivativo = 0;
130     }
131
132     // Debo compensar? -----
133     Salida = Proporcional + Integral + Derivativo;
134     Compensacion = 0;
135     if (LimitaSalida && CompensaIntegral) {
136         // Si no hay limite de saturacion a la salida, no hay nada que compensar...
137         if (Salida > SalidaMax) {
138             // Debo saturar la salida porque supera el maximo...
139             Compensacion = Salida - SalidaMax;
140         }
141         if (Salida < SalidaMin) {
142             // Debo saturar la salida porque esta por debajo del minimo...
143             Compensacion = Salida - SalidaMin;
144         }
145     }
146 }
```

## Trabajo de Laboratorio 2: Control Lineal de un Ascensor

```
147 // INTEGRAL -----
148 if (TiempoAnterior > 0 && Ti != 0) {
149     // Cumplidas las condiciones para integrar: (Si Compensacion==0, no va a compensar nada...)
150     Integral += (Kp * (ERROR + ErrorAnterior) - (Compensacion + CompensacionAnterior))
151         * (TiempoActual - TiempoAnterior)
152     / (2 * Ti * MILLON);
153     if (LimitaSalida) {
154         // Debo saturar la integral:
155         // (se supone que esto solo podria pasar si cambio los parametros de integracion)
156         Integral = min(Integral, SalidaMax);
157         Integral = max(Integral, SalidaMin);
158     }
159 }
160
161 // Termina componente integral -----
162
163 // Calculo final completo:
164 Salida = Proporcional + Integral + Derivativo;
165
166 if (LimitaSalida) {
167     // Debo saturar la salida:
168     // (se supone que esto solo podria pasar si cambio los parametros de integracion)
169     Salida = min(Salida, SalidaMax);
170     Salida = max(Salida, SalidaMin);
171 }
172 TiempoAnterior = TiempoActual;
173 ErrorAnterior = ERROR;
174 CompensacionAnterior = Compensacion;
175 return Salida;
176 // Termina funcion PID -----
177 }
178 //-----
179
180 float controlPID::ObtenerIntegral()
181 {
182     return Integral;
183 }
184 //-----
185
186 float controlPID::ObtenerProporcional()
187 {
188     return Proporcional;
189 }
190 //-----
191
192 float controlPID::ObtenerDerivativo()
193 {
194     return Derivativo;
195 }
196 //-----
197
198 float controlPID::ObtenerSalida()
199 {
200     return Salida;
201 }
202 //-----
203
204 float controlPID::ObtenerCompensacion()
205 {
206     return Compensacion;
207 }
208 //-----
209
210 void controlPID::Apagar()
211 {
212     TiempoAnterior = 0;
213     ErrorAnterior = 0;
214     CompensacionAnterior = 0;
215     Integral = 0;
216     Proporcional = 0;
217     Derivativo = 0;
218 }
```

## Bibliografía

PWM

<https://solectroshop.com/es/blog/que-es-pwm-y-como-usarlo--n38>

Kuo - "Sistemas de control automático - 7ed"

Codigo para control PID

[https://github.com/sca-undav/Control\\_PID](https://github.com/sca-undav/Control_PID)